

Cycles-limites et modèles de croissance goodwiniens

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Ce chapitre poursuit l'étude des modèles de croissance hétérodoxes d'inspiration goodwinienne. On a vu que le modèle initial de lutte des classes pour le partage du revenu national générait des courbes fermées dans le plan des phases et par conséquent des trajectoires économiques marquées par des cycles perpétuels. Cette propriété souffre du défaut mathématique d'instabilité structurelle : des modifications mineures dans la forme structurelle du modèle dynamique peuvent se traduire par la perte de cyclicité des trajectoires solutions. Samuelson, grand argentier de l'orthodoxie, a très vite repéré ce problème. Shah et Desai (voir section 3 ci-dessous), pourtant proches du paradigme goodwinien, sont parvenus à la même conclusion en greffant un mécanisme de progrès technique endogène améliorant la productivité des travailleurs sans effet sur la détermination des salaires.

La réaction des post-goodwiniens au défaut de stabilité structurelle a consisté à mettre au point des modèles dynamiques générant des cycles-limites dans l'espace des phases. Deux voies formelles ont été explorées : application du théorème de Poincaré-Bendixon et utilisation de la bifurcation super critique de Hopf. Ce chapitre aborde ces questions difficiles à travers les modèles de Jarsulic (1986), de Skott (1988) et de van der Ploeg (1987). L'accent est moins mis sur leurs propriétés mathématiques spécifiques que sur la possibilité d'illustrer ces propriétés au moyen des ressources graphiques offertes par Maple.

▼ Modèle de Jarsulic (1986)

Marc Jarsulic signe en 1986 un article intitulé "Growth Cycles in a Classical-Keynesian Model" qui est souvent considéré comme le prototype emblématique d'un modèle goodwinien satisfaisant le critère de la stabilité structurelle.

Dans la tradition goodwinienne, il s'agit de mettre en évidence un phénomène de croissance avec fluctuations endogènes provoquées implicitement par les conflits de répartition. Par rapport au travail pionnier de Goodwin, l'amélioration consiste à générer des cycles limites en greffant sur le mécanisme marxien de formation des salaires un mécanisme keynésien d'adaptation de l'offre à la demande. La sphère de la production est caractérisée par la complémentarité des facteurs capital et travail. Le produit et l'emploi sont fixés par le principe de la demande effective. L'investissement s'égalise à l'épargne globale, qui est le simple produit du revenu global par la propension moyenne à épargner. Le taux d'accumulation du capital désiré par les firmes dépend de la profitabilité. Les ajustements macroéconomiques se font par les quantités et non par des variations de prix.

▼ Forme structurelle

On adopte les conventions d'écriture suivantes :

1. Y , K , I , S , w , L et N sont respectivement le revenu, le capital, l'investissement, l'épargne, le salaire moyen, la population active occupée et la population active totale.
2. le taux d'emploi est noté $\varepsilon = \frac{L}{N}$.
3. la part des salaires dans le revenu est $\omega = \frac{wL}{Y}$.
4. le taux d'accumulation du capital est noté $g = \frac{I}{K}$.
5. la propension à épargner est notée $s = \frac{S}{Y}$.

La forme structurelle consiste en un système de deux équations différentielles ordinaire (EDO) dont les variables d'état sont g et ω . Chaque EDO est posée plutôt que déduite d'un système d'équations comptables, de définition et de comportement retraçant le circuit macroéconomique.

Evolution du taux d'accumulation du capital g

Chez Jarsulic, le rapport $\frac{Y}{K}$ n'est pas une donnée technologique donnant la productivité apparente du capital. Il mesure le taux d'utilisation des capacités de production. Dans la tradition des modèles de croissance keynésiens fermés et sans Etat où le revenu global est déterminé par le principe de la demande effective, la conséquence de l'égalité entre l'épargne et l'investissement est l'équation $\frac{Y}{K} = \frac{g}{s}$.

Le taux d'accumulation g est ici une variable endogène qui dépend de la profitabilité à deux niveaux. Première idée : si le taux d'accumulation est très faible, il recommencera à remonter dès que la part des salaires aura elle-même atteint un niveau très faible (cet effet veut légitimer la reprise quand l'économie est au fond du trou). Seconde idée : il existe un taux d'accumulation maximum, impossible à dépasser même si les "travailleurs pouvaient vivre de l'air du temps".

Pour Jarsulic, ces deux effets sont captés dans le taux de croissance du taux d'accumulation du capital par l'EDO suivante :

$$\frac{\dot{g}}{g} = \frac{a + b g - c g^2}{d + e \omega} - f \text{ avec } a, b, c, d, e, f > 0 \text{ et } a > d f$$

Le moins qu'on puisse dire est que cette expression ne va pas de soi. Il n'est pas inutile de commencer par visualiser les mécanismes sous-jacents.

> restart;


```
Fg:=(x,y,a,b,c,d,e,f)->(a+b*x-c*x^2)/(d+e*y)-f;#fonction
générale définissant le taux d'accumulation à partir des
variables endogènes x et y et des paramètres
Fgs:=unapply(Fg(x,y,0.3,1,40,1,0.75,0.11),(x,y));#fonction
définissant le taux d'accumulation à partir des paramètres
fournis par Jarsulic
```

$$Fg := (x, y, a, b, c, d, e, f) \rightarrow \frac{a + b x - c x^2}{d + e y} - f$$

$$Fgs := (x, y) \rightarrow \frac{0.3 + x - 40 x^2}{1 + 0.75 y} - 0.11 \quad (1.1.1.1)$$


Le graphique suivant donne la surface représentative de la fonction **Fgs** dans le plan $Og\omega$. On vérifie que **Fgs** a les propriétés requises : pour un taux d'accumulation donnée, son taux de croissance est d'autant plus élevé que la part des salaires est faible; pour un taux de salaires donné, le taux de variation du taux d'accumulation est une fonction parabolique du taux d'accumulation qui est croissante pour des taux d'accumulation faibles, admet un maximum, puis décroît pour des taux d'accumulation de plus en plus élevés.

```
> plot3d(Fgs(g,omega),g=0..0.1,omega=0..1,axes=normal,
orientation=[-61,53,179]);#représentation graphique de FGs
```




Détaillons chaque effet au moyen de coupes. D'abord, les coupes suivant ω montrent clairement que plus la part des salaires est élevée, plus l'effet du taux d'accumulation sur sa croissance est faible. Dans tous les cas, la courbe admet un maximum.

```
> plot([Fgs(g,0.5),Fgs(g,0.7),Fgs(g,0.8),Fgs(g,0.9)],g=0.09,color=[black,greyscale,red,blue]);#4 coupes en maintenant la part des salaires constante
```



A présent, on fixe le taux d'accumulation à différents niveaux et on examine l'impact de la part des salaires. Même quand la part des salaires est nulle, le taux de croissance du taux d'accumulation est bornée. Mais plus le taux d'accumulation est élevé, plus la courbe est translatée vers le bas.

```
> plot([Fgs(0.025,omega),Fgs(0.05,omega),Fgs(0.075,omega)],
       omega=0..1,color=[black,gre,red]);
```



▼ Evolution de la part des salaires ω

La part des salaires dérive d'un mécanisme goodwinien implicite dans lequel le taux d'emploi est une fonction linéaire croissante du taux d'accumulation du capital. Plus l'accumulation est forte, plus les firmes embauchent et la hausse du taux d'emploi favorise les hausses salariales. Recul du chômage et augmentation du salaire moyen se conjuguent pour accroître la part des salaires dans le revenu. On a ainsi :

$$\frac{\dot{\omega}}{\omega} = (1 - \omega)(hg - j) \text{ avec } h, j > 0$$


La présence du facteur $1 - \omega$ s'explique par le fait qu'on contraint la part des salaires à rester dans l'intervalle $[0; 1]$: si ω tend vers 0, la part des salaires ne bouge plus; si ω tend vers 1, elle se stabilise.

On visualise la surface correspondant à la dérivée de la part des salaires, soit la fonction définie par $f(g, \omega) = \omega(1 - \omega)(hg - j)$

```
> Fo:=(x,y,h,j)->y*(1-y)*(h*x-j);#fonction générale donnant
la variation de la part des salaires en fonction des 2
variables endogènes et des 2 paramètres
Fos:=unapply(Fo(x,y,2,0.28),(x,y));#fonction avec
paramètres fixés
plot3d(Fos(g,omega),omega=0..1,g=0..0.3,axes=box,
orientation=[-120,76]);#surface représentative de Fos
```

$$Fo := (x, y, h, j) \rightarrow y(1-y)(hx-j)$$


$$Fos := (x, y) \rightarrow y(1-y)(2x-0.28)$$



A nouveau, les coupes sont plus parlantes.


Première coupe : le taux d'accumulation est constant (pour différents niveaux). Il détermine la concavité de la parabole. Pour des niveaux d'accumulation faibles, la courbe est convexe et la croissance de la part des salaires passe par un minimum.. Pour des taux d'accumulation élevés, la courbe est concave et la vitesse de la part des salaires passe par un maximum.

```
> plot([Fos(0,omega),Fos(0.1,omega),Fos(0.2,omega),Fos(0.3,omega)],omega=0..1,color=[black,red,blue]);
```



Deuxième coupe : la part des salaires est constante (pour différents niveaux). De faibles taux d'accumulation s'accompagnent d'une baisse de la part des salaires. Une accumulation élevée accroît la part des salaires. Les effets sont d'autant plus marqués que la part des salaires est faible.

```
> plot([Fos(g,0.5),Fos(g,0.6),Fos(g,0.7),Fos(g,0.8)],g=0.0.3,color=[black,greym,red,blue]);
```



▼ *Système différentiel*

Le coeur du modèle de Jarsulic est le système différentiel non linéaire suivant :

$$(SDJ) \quad \begin{cases} \dot{g}(t) = g(t) \left[\frac{a + b g(t) - c g(t)^2}{d + e \omega(t)} - f \right] \\ \dot{\omega}(t) = \omega(t) (1 - \omega(t)) (h g(t) - j) \end{cases}$$

> `edo1:=diff(g(t),t)=g(t)*Fg(g(t),omega(t),a,b,c,d,e,f);`
`edo2:=diff(omega(t),t)=Fo(g(t),omega(t),h,j);`

$$edo1 := \frac{d}{dt} g(t) = g(t) \left(\frac{a + b g(t) - c g(t)^2}{d + e \omega(t)} - f \right)$$

$$edo2 := \frac{d}{dt} \omega(t) = \omega(t) (1 - \omega(t)) (h g(t) - j)$$

(1.1.3.1)

Il faut faire le deuil de la résolution explicite. Au bout d'une heure, Maple se contente de générer une équation différentielle (insoluble à ce niveau de généralité) du second degré en ω .

```

> infolevel[dsolve]:=2;#demande d'informations sur les
étapes de la résolution
dsolve({edo1,edo2},{g(t),omega(t)});
infoleveldsolve:=2
-> Solving each unknown as a function of the next ones
using the order: [g(t), omega(t)]
-> Calling odsolve with the ODE diff(diff(y(x) x) x) =
(4*(diff(y(x) x))*y(x)^3*j*b*h-2*(diff(y(x) x))*y(x)^2*
j*b*h-2*(diff(y(x) x))*y(x)^4*j*b*h-(diff(y(x) x))*a*y
(x)^2*h^2-3*f*e*y(x)^5*h^3+3*f*e*y(x)^6*h^3-f*e*y(x)^7*
h^3+f*d*y(x)^3*h^3-3*f*d*y(x)^4*h^3+3*f*d*y(x)^5*h^3-f*
d*y(x)^6*h^3-(diff(y(x) x))^2*d*h^2*y(x)+3*(diff(y(x) x)
)^2*d*h^2*y(x)^2-(diff(y(x) x))^2*e*y(x)^2*h^2+3*(diff(y
(x) x))^2*e*y(x)^3*h^2-2*y(x)^3*(diff(y(x) x))^2*d*h^2
-2*y(x)^4*(diff(y(x) x))^2*e*h^2+f*e*y(x)^4*h^3+2*(diff
(y(x) x))*a*y(x)^3*h^2+(diff(y(x) x))^2*b*y(x)^2*h-3*
(diff(y(x) x))^2*c*y(x)^2*j-(diff(y(x) x))^2*b*y(x)*h+3*
(diff(y(x) x))^2*c*y(x)*j-6*(diff(y(x) x))*y(x)^3*j^2*
c+3*(diff(y(x) x))*y(x)^2*j^2*c+3*(diff(y(x) x))*y(x)^4*
j^2*c-y(x)^3*j*a*h^2-3*y(x)^5*j*a*h^2+3*y(x)^4*j*a*
h^2+3*y(x)^4*j^2*b*h-y(x)^3*j^2*b*h-3*y(x)^5*j^2*b*h+y
(x)^6*j*a*h^2+y(x)^6*j^2*b*h-3*y(x)^4*j^3*c+y(x)^3*j^3*
c+3*y(x)^5*j^3*c-y(x)^6*j^3*c+c*(diff(y(x) x))^3-(diff(y
(x) x))*a*y(x)^4*h^2)/(-y(x)^2*d*h^2+2*y(x)^3*d*h^2-y(x)
^4*d*h^2-y(x)^3*e*h^2+2*y(x)^4*e*h^2-y(x)^5*e*h^2) y(x)
singsol = none
Methods for second order ODEs:
--- Trying classification methods ---
trying 2nd order Liouville
trying 2nd order WeierstrassP
trying 2nd order JacobisN
differential order: 2; trying a linearization to 3rd
order
trying 2nd order ODE linearizable_by_differentiation
trying 2nd order, 2 integrating factors of the form mu
(x,y)
trying differential order: 2; missing variables
-> Computing symmetries using: way = 3
-> Computing symmetries using: way = exp_sym
-> Calling odsolve with the ODE (diff(_b(_a) _a))*_b(_a)
+(-3*_a^4*j^3*c+_a^3*j^3*c+3*_a^5*j^3*c-_a^6*j^3*c-3*f*
e*_a^5*h^3+3*f*e*_a^6*h^3-f*e*_a^7*h^3+f*d*_a^3*h^3-3*f*
d*_a^4*h^3+3*f*d*_a^5*h^3-f*d*_a^6*h^3+f*e*_a^4*h^3-
_a^3*j*a*h^2-3*_a^5*j*a*h^2+3*_a^4*j*a*h^2+3*_a^4*j^2*b*
h-_a^3*j^2*b*h-3*_a^5*j^2*b*h+_a^6*j*a*h^2+_a^6*j^2*b*h-
_b(_a)*a*_a^2*h^2-_b(_a)^2*d*h^2*_a+3*_b(_a)^2*d*h^2*
_a^2-_b(_a)^2*e*_a^2*h^2+3*_b(_a)^2*e*_a^3*h^2-2*_a^3*_b
(_a)^2*d*h^2-2*_a^4*_b(_a)^2*e*h^2+2*_b(_a)*a*_a^3*h^2+
_b(_a)^2*b*_a^2*h-3*_b(_a)^2*c*_a^2*j-_b(_a)^2*b*_a*h+3*
_b(_a)^2*c*_a*j-6*_b(_a)*_a^3*j^2*c+3*_b(_a)*_a^2*j^2*
c+3*_b(_a)*_a^4*j^2*c-_b(_a)*a*_a^4*h^2+4*_b(_a)*_a^3*j*
b*h-2*_b(_a)*_a^2*j*b*h-2*_b(_a)*_a^4*j*b*h+c*_b(_a)^3)/
(h^2*_a^2*(d-2*d*_a+d*_a^2+_a*e-2*_a^2*e+_a^3*e)) = 0 _b
(_a)
*** Sublevel 2 ***
Methods for first order ODEs:

```

```

--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
trying Abel
Looking for potential symmetries
Looking for potential symmetries
Looking for potential symmetries
trying inverse_Riccati
trying an equivalence to an Abel ODE
trying 1st order ODE linearizable_by_differentiation
--- Trying Lie symmetry methods, 1st order ---
-> Computing symmetries using: way = 3
-> Computing symmetries using: way = 4
-> Computing symmetries using: way = 2
trying symmetry patterns for 1st order ODEs
-> trying a symmetry pattern of the form [F(x)*G(y),
0]
-> trying a symmetry pattern of the form [0, F(x)*G
(y)]
-> trying symmetry patterns of the forms [F(x),G(y)]
and [G(y),F(x)]
-> trying a symmetry pattern of the form [F(x),G(x)]
-> trying a symmetry pattern of the form [F(y),G(y)]
-> trying a symmetry pattern of the form [F(x)+G(y),
0]
-> trying a symmetry pattern of the form [0, F(x)+G
(y)]
-> trying a symmetry pattern of the form [F(x),G(x)*
y+H(x)]
-> trying a symmetry pattern of conformal type
trying differential order: 2; exact nonlinear
trying 2nd order, integrating factor of the form mu(x,y)
--- Trying Lie symmetry methods, 2nd order ---
-> Computing symmetries using: way = 3
-> Computing symmetries using: way = 5
-> Computing symmetries using: way = formal

```

$$\left[\left\{ \omega(t) = 0 \right\}, \left\{ g(t) = \text{RootOf} \left(t - \left(\int^{-Z} \left(-\frac{d}{c_a^3 + fd - a a - b_a^2} \right) d_a \right) + _Cl \right) \right\} \right], \left[\left\{ \omega(t) = 1 \right\}, \left\{ g(t) = \text{RootOf} \left(t - \left(\int^{-Z} \left(-\frac{d+e}{c_a^3 + fd - a a - b_a^2 + fe} \right) d_a \right) + _Cl \right) \right\} \right], \left[\left\{ \frac{d^2}{dt^2} \omega(t) \right. \right.
\tag{1.1.3.2}$$

$$\begin{aligned}
&= \left(4 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^3 j b h - 2 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^2 j b h - 2 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^4 j b h\right. \\
&- \left(\frac{d}{dt} \omega(t)\right) a \omega(t)^4 h^2 - \left(\frac{d}{dt} \omega(t)\right) a \omega(t)^2 h^2 - 3 f e \omega(t)^5 h^3 \\
&+ 3 f e \omega(t)^6 h^3 - f e \omega(t)^7 h^3 + f d \omega(t)^3 h^3 - 3 f d \omega(t)^4 h^3 + 3 f d \omega(t)^5 h^3 \\
&- f d \omega(t)^6 h^3 - \left(\frac{d}{dt} \omega(t)\right)^2 d h^2 \omega(t) + 3 \left(\frac{d}{dt} \omega(t)\right)^2 d h^2 \omega(t)^2 \\
&- \left(\frac{d}{dt} \omega(t)\right)^2 e \omega(t)^2 h^2 + 3 \left(\frac{d}{dt} \omega(t)\right)^2 e \omega(t)^3 h^2 \\
&- 2 \omega(t)^3 \left(\frac{d}{dt} \omega(t)\right)^2 d h^2 - 2 \omega(t)^4 \left(\frac{d}{dt} \omega(t)\right)^2 e h^2 + f e \omega(t)^4 h^3 \\
&+ 2 \left(\frac{d}{dt} \omega(t)\right) a \omega(t)^3 h^2 + \left(\frac{d}{dt} \omega(t)\right)^2 b \omega(t)^2 h \\
&- 3 \left(\frac{d}{dt} \omega(t)\right)^2 c \omega(t)^2 j - \left(\frac{d}{dt} \omega(t)\right)^2 b \omega(t) h + 3 \left(\frac{d}{dt} \omega(t)\right)^2 c \omega(t) j \\
&- 6 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^3 j^2 c + 3 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^2 j^2 c \\
&+ 3 \left(\frac{d}{dt} \omega(t)\right) \omega(t)^4 j^2 c - \omega(t)^3 j a h^2 - 3 \omega(t)^5 j a h^2 + 3 \omega(t)^4 j a h^2 \\
&+ 3 \omega(t)^4 j^2 b h - \omega(t)^3 j^2 b h - 3 \omega(t)^5 j^2 b h + \omega(t)^6 j a h^2 + \omega(t)^6 j^2 b h \\
&- 3 \omega(t)^4 j^3 c + \omega(t)^3 j^3 c + 3 \omega(t)^5 j^3 c - \omega(t)^6 j^3 c + c \left(\frac{d}{dt} \omega(t)\right)^3 \Big) / (\\
&- \omega(t)^2 d h^2 + 2 \omega(t)^3 d h^2 - \omega(t)^4 d h^2 - \omega(t)^3 e h^2 + 2 \omega(t)^4 e h^2
\end{aligned}$$

$$-\omega(t)^5 e h^2), \left\{ g(t) = \frac{-\left(\frac{d}{dt} \omega(t)\right) - \omega(t) j + \omega(t)^2 j}{-\omega(t) h + \omega(t)^2 h} \right\}$$

On mène donc une étude qualitative du système en commençant par repérer les équilibres dynamiques puis en cherchant les propriétés locales.

Recherche des équilibres dynamiques

Les équilibres dynamiques, encore appelés points de repos ou états stationnaires, sont les fonctions constantes vérifiant (SDJ), c'est à dire ici les réels annulant le système d'équations formé par $\mathbf{g}^* \mathbf{Fg} = 0$ et $\mathbf{Fo} = 0$.

```
> eq1:=g*Fg(g,omega,a,b,c,d,e,f)=0;
eq2:=Fo(g,omega,h,j)=0;
soleq:=solve({eq1,eq2},[g,omega]);
```

$$eq1 := g \left(\frac{a + b g - c g^2}{d + e \omega} - f \right) = 0$$

$$eq2 := \omega (1 - \omega) (h g - j) = 0$$

$$soleq := \left[[g=0, \omega=0], [g=0, \omega=1], \left[g = \frac{j}{h}, \omega = \frac{a h^2 + b j h - c j^2 - f d h^2}{h^2 f e} \right], [g = \text{RootOf}(-a - b _Z + c _Z^2 + f d), \omega=0], [g = \text{RootOf}(f e - a - b _Z + c _Z^2 + f d), \omega=1] \right] \quad (1.2.1)$$

Formellement, le système admet 7 points d'équilibre, en fait 5 si on se souvient qu'on essaye de résoudre un problème économique et qu'on se restreint au quadrant positif du plan des phases, (quel sens donner à une désaccumulation perpétuelle du capital?) et même 4 s'il n'y a pas d'intersection entre l'isocline $\dot{g} = 0$ pour $g \neq 0$ et $\omega = 1$. Pour visualiser le problème, cherchons les isoclines.

L'isocline $\dot{\omega} = 0$ correspond à $\omega = 0$ ou $\omega = 1$ ou $g = \frac{j}{h}$. L'isocline $\dot{g} = 0$ correspond à $g = 0$ ou

$$\frac{a + b g - c g^2}{d + e \omega} - f = 0. \text{ Cette dernière équation donne une relation parabolique entre } \omega \text{ et } g.$$


Manifestement, Jarsulic suppose implicitement que le maximum atteint par la fonction parabolique $\omega = \varphi(g)$ est strictement inférieur à 1, ce qui, avec la condition de départ $a > df$, restreint la liberté de choix pour les paramètres. Pour la représentation des isoclines dans le plan des phases, nous proposons le jeu de paramètres suivant, qui satisfait cette condition (au passage, signalons que le jeu de paramètres proposés par Jarsulic ne convient pas!):

$$a = 0.1225; b = 0.7875; c = 4.5; d = 1; e = 0.75; f = 0.1; h = 2; j = 0.12$$

Bien d'autres calibrations sont évidemment possibles.

```
> g1:=plot(solve(Fg(g,omega,0.1225,0.7875,4.5,1,0.75,0.1)=0,
omega),g=0..0.2,thickness=2):
g2:=plot([[0,0],[0,1.1]],0..0.22,0..1.2,style=line,thickness=
2):#deuxième isocline g'=0
g3:=plot([[0,0],[0.2,0]],0..0.22,0..1.2,style=line,thickness=
2):#1° isocline omega'=0
g4:=plot([[0,1],[0.20,1]],0..0.22,0..1.2,style=line,
thickness=2):#2° isocline omega'=0
g5:=plot([[0.12/2,0],[0.12/2,1.1]],0..0.22,0..1.2,style=line,
thickness=2):#3° isocline omega'=0
```

```
with(plots):
g_isoc:=display(g1,g2,g3,g4,g5):g_isoc;#tracé des isoclines
dans le plan des phases
```



On repère 4 points d'intersection, notés E_1 , E_2 , E_3 et E_4 , entre les isoclines $\dot{g} = 0$ et $\dot{\omega} = 0$. Ces équilibres dynamiques ne sont pas tous intéressants : E_2 , E_3 et E_4 correspondent à des situations économiques "extrêmes" où la part des salaires dans le revenu national est nulle ou égale à 1 (auquel cas, la part des profits est nulle!). En définitive, seul l'équilibre non trivial

$E_1 = \left(g = \frac{j}{h}, \omega = \frac{a h^2 + b j h - c j^2 - f d h^2}{h^2 f e} \right)$ présente un réel intérêt économique et c'est celui

sur lequel vont porter désormais tous les efforts. Pour la suite, on notera g_{eq} et ω_{eq} les coordonnées du point **E1**.

▼ Propriétés locales de l'équilibre non trivial

On cherche maintenant à connaître le comportement de la dynamique au voisinage du point d'équilibre non trivial. Il faut explorer la jacobienne de (SDJ) en ce point.

L'expression de la jacobienne est aisément calculable :

```
> Jac:=Matrix(2,2,[[diff(g*Fg(g,omega,a,b,c,d,e,f),g),diff(g*Fg
```

$$\begin{aligned}
 & (g, \omega, a, b, c, d, e, f), \omega), [diff(Fo(g, \omega, h, j), g), diff \\
 & (Fo(g, \omega, h, j), \omega)]]; \#matrice jacobienne en tout point \\
 Jac := & \begin{bmatrix} \frac{a + b g - c g^2}{d + e \omega} - f + \frac{g (b - 2 c g)}{d + e \omega} & - \frac{g (a + b g - c g^2) e}{(d + e \omega)^2} \\ \omega (1 - \omega) h & (1 - \omega) (h g - j) - \omega (h g - j) \end{bmatrix} \quad (1.3.1)
 \end{aligned}$$

En revanche, il faut utiliser subtilement Maple pour donner une expression maniable de la jacobienne à l'équilibre E_1 . On préférera avancer pas à pas, en tenant compte des particularités des composantes de cet équilibre, à savoir : $\frac{a + b g - c g^2}{d + e \omega} - f = 0$ et $h g - j = 0$.

```

> b(1,1):=g*(b-2*c*g)/(d+e*omega_eq);b(1,1):=subs(g=g_eq,b(1,1));
b(1,2):=-g*f*e/(d+e*omega_eq);b(1,2):=subs({g=g_eq,omega=omega_eq},b(1,2));
b(2,1):=subs(omega=omega_eq,Jac[2,1]);
b(2,2):=0;
Jac_eq:=Matrix(2,2,[[b(1,1),b(1,2)],[b(2,1),b(2,2)]]);
#jacobienne au point d'équilibre

```

$$\begin{aligned}
 b(1,1) &:= \frac{g (b - 2 c g)}{d + e \omega_{eq}} \\
 b(1,1) &:= \frac{g_{eq} (b - 2 c g_{eq})}{d + e \omega_{eq}} \\
 b(1,2) &:= - \frac{g f e}{d + e \omega_{eq}} \\
 b(1,2) &:= - \frac{g_{eq} f e}{d + e \omega_{eq}} \\
 b(2,1) &:= \omega_{eq} (1 - \omega_{eq}) h \\
 b(2,2) &:= 0
 \end{aligned}$$

$$Jac_{eq} := \begin{bmatrix} \frac{g_{eq} (b - 2 c g_{eq})}{d + e \omega_{eq}} & - \frac{g_{eq} f e}{d + e \omega_{eq}} \\ \omega_{eq} (1 - \omega_{eq}) h & 0 \end{bmatrix} \quad (1.3.2)$$

```

> with(LinearAlgebra):
simplify(Determinant(Jac_eq));#calcul du déterminant de Jac_eq
simplify(Trace(Jac_eq));#calcul de la trace de Jac_eq

```

$$\begin{aligned}
 & - \frac{g_{eq} f e \omega_{eq} (-1 + \omega_{eq}) h}{d + e \omega_{eq}} \\
 & \frac{g_{eq} (b - 2 c g_{eq})}{d + e \omega_{eq}} \quad (1.3.3)
 \end{aligned}$$

En tenant compte du fait que le taux d'accumulation g_{eq} est positif, que la part des salaires ω_{eq} est comprise entre 0 et 1 et que tous les paramètres sont positifs, le déterminant est positif et les 2 valeurs propres de la jacobienne Jac_{eq} sont de même signe. Par contre, le signe de la trace n'est pas déterminé et dépend crucialement de la valeur prise par le taux d'accumulation d'équilibre $g_{eq} = \frac{j}{h}$ par rapport au taux d'accumulation g qui maximise l'isocline

$\dot{g} = 0$, c'est à dire $g = \frac{b}{2c}$. On est donc conduit à distinguer trois cas de figure :

1°) Si $\frac{j}{h} < \frac{b}{2c}$, la trace est positive et l'équilibre est répulsif (localement instable).

2°) Si $\frac{j}{h} > \frac{b}{2c}$, la trace est négative et l'équilibre est attracteur (localement stable).

3°) Si $\frac{j}{h} = \frac{b}{2c}$, la trace est nulle et l'équilibre est un centre.

▼ Analyse qualitative du cas stable

On illustre facilement la stabilité quand $\frac{j}{h} > \frac{b}{2c}$.

Modifions simplement le paramètre j dans le jeu de paramètres utilisé plus haut pour visualiser les isoclines :

$a = 0.1225; b = 0.7875; c = 4.5; d = 1; e = 0.75; f = 0.1; h = 2; j = 0.28$ (au lieu de $j = 0.12$).


On a bien $\frac{j}{h} = 0.14 > \frac{b}{2c} = 0.0875$.

La commande **display** du paquetage **plots** permet de tracer dans le plan des phases les isoclines ainsi que le champ directionnel et quelques orbites obtenus à l'aide de la commande **DEplot** du paquetage **DEtools**.

```
> g6:=plot([[0.28/2,0],[0.28/2,1.1]],0..0.22,0..1.2,style=line,
thickness=2):#isocline omega'=0 d'équation g=j/h
edo1_s:=diff(g(t),t)=g(t)*Fg(g(t),omega(t),0.1225,0.7875,4.5,
1,0.75,0.1);#EDO en g
edo2_s:=diff(omega(t),t)=Fo(g(t),omega(t),2,0.28);#EDO en
omega
with(DEtools):#chargement du paquetage
g_orbs:=DEplot({edo1_s,edo2_s},{g(t),omega(t)},t=0..150,g=0.
.0.20,omega=0..1,[[g(0)=0.17,omega(0)=0.05],[g(0)=0.08,omega
(0)=0.4],[g(0)=0.12,omega(0)=0.9],[g(0)=0.17,omega(0)=0.85]],
linecolor=black,thickness=1):#résolution numérique avec 4
conditions initiales
display(g1,g2,g3,g4,g6,g_orbs);
```

$$edo1_s := \frac{d}{dt} g(t) = g(t) \left(\frac{0.1225 + 0.7875 g(t) - 4.5 g(t)^2}{1 + 0.75 \omega(t)} - 0.1 \right)$$

$$edo2_s := \frac{d}{dt} \omega(t) = \omega(t) (1 - \omega(t)) (2 g(t) - 0.28)$$



La convergence asymptotique avec fluctuations vers l'équilibre est bien mise en valeur.

▼ Analyse qualitative du cas instable

Tout se joue ici. Le cas d'instabilité va mettre en évidence l'apparition d'un cycle-limite.

On procède de la même manière que dans la section précédente, mais avec le jeu de paramètres suivant (initialement utilisé dans la section "recherche des équilibres") :


$a = 0.1225; b = 0.7875; c = 4.5; d = 1; e = 0.75; f = 0.1; h = 2; j = 0.14$

qui vérifie la condition d'instabilité locale : $\frac{j}{h} = 0.07 < \frac{b}{2c} = 0.0875$.

```
> edo1_i:=diff(g(t),t)=g(t)*Fg(g(t),omega(t),0.1225,0.7875,4.5,
1,0.75,0.1);#EDO en g
edo2_i:=diff(omega(t),t)=Fo(g(t),omega(t),2,0.14);#EDO en
omega
g_orbi:=DEplot({edo1_i,edo2_i},{g(t),omega(t)},t=0..600,g=0.
.0.2,omega=0..1,[[g(0)=0.068,omega(0)=0.7],[g(0)=0.2,omega(0)
=0.6]],stepsize=0.01,linecolor=black,thickness=1):#résolution
numérique avec 2 conditions initiales
display(g1,g2,g3,g4,g5,g_orbi);
```

$$edo1_i := \frac{d}{dt} g(t) = g(t) \left(\frac{0.1225 + 0.7875 g(t) - 4.5 g(t)^2}{1 + 0.75 \omega(t)} - 0.1 \right)$$

$$\text{edo2}_i := \frac{d}{dt} \omega(t) = \omega(t) (1 - \omega(t)) (2g(t) - 0.14)$$



La résolution demande environ 30 secondes pour avoir des orbites précises!!! Mais cela vaut la peine d'attendre.

Le graphique met nettement en évidence le fait que l'équilibre est localement instable (les orbites démarrant près de l'équilibre divergent). Mais on engrange un résultat supplémentaire grâce à Maple : une orbite démarrant "loin" de l'équilibre s'en rapproche. On pressent un cycle-limite. Pour le démontrer, Jarsulic propose un raisonnement intuitif proche de la démonstration du théorème de Poincaré-Bendixon.

Démarrer en P_2 place sur une orbite divergente conformément aux conclusions de l'étude locale.


Par contre, démarrer en P_1 place sur une orbite qui rapproche de l'équilibre. Le flot se dirige d'abord vers le nord-ouest sans atteindre l'isocline $\omega = 1$, franchit l'isocline $\dot{\omega} = 0$ pour se diriger vers le sud-ouest, franchit l'isocline $\dot{g} = 0$ pour se diriger vers le sud-est sans atteindre l'isocline $\omega = 0$, franchit l'isocline $\dot{\omega} = 0$ pour se diriger vers le nord-est, franchit l'isocline $\dot{g} = 0$ et revient dans la zone de direction nord-ouest. L'orbite ne peut pas se croiser avec elle-même en vertu du théorème de Cauchy. Elle se "replie" et converge nécessairement d'une manière spiralee vers un cycle-limite.

Notons enfin que le théorème de la bifurcation de Hopf s'applique ici puisque le déterminant de la jacobienne est toujours positif (donc les valeurs propres ont le même signe) et que la trace

change de signe suivant que l'isocline $\dot{\omega} = 0$ est à gauche, sur ou à droite du maximum atteint par l'isocline $\dot{g} = 0$. Il suffit de choisir le paramètre j pour obtenir une bifurcation super-critique, à ceci près qu'on ne peut pas affirmer que le cycle-limite est unique. Les simulations avec Maple, comme cette animation présentée ci-dessous, tendent à montrer que c'est bien le cas, mais des simulations ne constituent pas une preuve.

Avec Maple, une bifurcation peut être visualisée au moyen d'une animation créée par l'option **insequence=true** dans la commande **display** qui va faire défiler la succession de 15 graphes réunissant chacun les isoclines, le champ directionnel et une orbite démarrant au même point, obtenue par balayage de valeurs pour le paramètre j (de $j = 0.28$ à $j = 0.14$ par pas de 0.01). Le "film" montre que l'orbite est d'abord convergente vers l'équilibre non trivial puis, une fois passée la valeur de bifurcation de j , l'orbite est attirée par un cycle-limite.

```
> display(
  seq(
    display(
      {g1,g2,g3,g4,
      plot([(0.28-0.01*k)/2,0],[0.28-0.01*k)/2,1.1]],0..0.22,0.
      .1.2,style=line,thickness=2),
      DEplot({diff(g(t),t)=g(t)*Fg(g(t),omega(t),0.1225,0.7875,4.5,
      1,0.75,0.1),diff(omega(t),t)=Fo(g(t),omega(t),2,0.28-0.01*k)}
      ,{g(t),omega(t)},t=0..500,g=0..0.2,omega=0..1,[g(0)=0.2,
      omega(0)=0.6]],linecolor=black,thickness=2,stepsize=0.1)}),
      k=0..14)
  ,insequence=true);#animation montrant la déformation de
  l'orbite démarrant au même point selon la valeur prise par le
  paramètre j
```



▼ Modèle de Skott (1989)

Peter Skott a publié en 1989 un article très astucieux dans lequel le croisement des intuitions goodwiniennes et kaldoriennes fonde un modèle dynamique non linéaire concluant à la persistance de cycles dans une économie en croissance. Deux sources de fluctuations endogènes se superposent

: les luttes sociales pour le partage du revenu et - fait nouveau dans la littérature goodwinienne - les conditions de travail; les déséquilibres ex ante sur le marché macroéconomique des biens et services.

Présentation générale

Dans la sphère de la production, le capital K et l'emploi efficace L sont fixés par des considérations comportementales et techniques.

L'accumulation du capital est donnée par l'équation de définition :

$$\frac{\dot{K}}{K} = \frac{I}{K} - \delta \text{ où } I \text{ désigne l'investissement et } \delta \text{ est le taux de dépréciation } \geq 0$$

Elle dépend donc des décisions d'investissements par les capitalistes dont on examinera la logique plus bas.

La productivité du travail efficace est supposée constante, de sorte que l'évolution de l'emploi se cale sur l'évolution de l'offre Y : $\frac{\dot{L}}{L} = \frac{\dot{Y}}{Y}$. En notant N la population active totale supposée croître

au taux démographique constant $n \geq 0$, le taux d'emploi $\varepsilon = \frac{L}{N}$ a pour croissance :

$$\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{L}}{L} - \frac{\dot{N}}{N} = \frac{\dot{Y}}{Y} - n$$

Le taux de croissance de l'output est déterminé par les firmes suivant la règle comportementale suivante :

$$\frac{\dot{Y}}{Y} = h(\pi, \varepsilon) \text{ avec } h_{\pi}' > 0 \text{ et } h_{\varepsilon}' < 0$$

L'influence positive de la part des profits dans le revenu national π s'explique par le fait que cette variable est ici un indicateur de la pression de la demande dans un univers de maximisation du profit à coûts marginaux constants. Plus la demande est élevée, plus les firmes ont intérêt à produire en fixant des prix supérieurs aux coûts marginaux afin d'accroître leurs profits marginaux, et partant les profits dont la part dans le revenu augmente. En conséquence, plus le niveau de la demande s'élève, plus la part des profits s'élève et plus le niveau de croissance désiré s'élève.

L'impact négatif du taux d'emploi ε sur la croissance provient de ce qu'il exprime, comme chez Goodwin, l'état des rapports sociaux ainsi que la disponibilité des travailleurs qualifiés. Dans les périodes de chômage faible, les salaires ont tendance à augmenter mais surtout les travailleurs remettent en cause la discipline de travail avec des méthodes de lutte contemporaines : turn over; ralentissement concerté des cadences (impliquant des coûts de surveillance et de mise en place de signaux); sabotage. Tous ces phénomènes découragent les décideurs dans leurs prévisions de croissance. A la limite, désirer moins de croissance revient à rechercher la paix sociale. A contrario, les périodes de chômage élevé cassent les mouvements sociaux et boostent les perspectives de croissance.

Sur le marché des biens et services, il y a déséquilibre ex ante entre l'épargne S et l'investissement I . A la manière de Kaldor, l'épargne en volume est le produit du revenu global Y par la propension à consommer s qui est une fonction croissante, notée g , de la part des profits dans le revenu π :

$$\frac{S}{Y} = g(\pi) \text{ avec } g'(\pi) > 0$$

L'investissement en volume désiré s'explique par deux effets. D'une part, la "profitabilité", mesurée par la part des profits π , influence positivement le taux d'investissement $\frac{I}{Y}$. D'autre

part, le degré d'utilisation du capital

$\sigma = \frac{Y}{K}$ exerce un effet accélérateur simple sur le taux

d'investissement. Ces deux effets jouent conjointement et on pose :

$$\frac{I}{Y} = f(\sigma, \pi) \text{ avec } f'_\sigma > 0 \text{ et } f'_\pi > 0$$

L'équilibre ex post ou comptable est obtenu quand l'égalité $I = S$ est vérifiée, c'est à dire si $g(\pi) = f(\sigma, \pi)$. Autant dire que l'ajustement ne se fait pas par variation du revenu comme chez Keynes, mais par un mouvement implicite des prix "à la Kaldor". Formellement, la part des profits est une fonction implicite du degré d'utilisation du capital. En effet, la différenciation de l'équation d'équilibre amène : $g'(\pi)d\pi = f'_\sigma(\sigma, \pi)d\sigma + f'_\pi(\sigma, \pi)d\pi$, soit encore :

$$\frac{d\pi}{d\sigma} = \frac{f'_\sigma(\sigma, \pi)}{g'(\pi) - f'_\pi(\sigma, \pi)}. \text{ Posons } \pi = \theta(\sigma). \text{ On a } \theta'(\sigma) > 0 \text{ si } g'(\pi) > f'_\pi(\sigma, \pi), \text{ hypothèse}$$

qu'on retiendra et qui signifie que l'épargne est plus réactive que l'investissement à une augmentation du profit.

Forme structurelle

La forme structurelle du modèle de Skott comprend 8 équations :

$$(1) \quad \frac{\dot{K}}{K} = \frac{I}{K} - \delta \text{ avec } \delta \geq 0$$

$$(2) \quad \frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{Y}}{Y} - n \text{ avec } n \geq 0$$

$$(3) \quad \frac{\dot{Y}}{Y} = h(\pi, \varepsilon) \text{ avec } h'_\pi > 0 \text{ et } h'_\varepsilon < 0$$

$$(4) \quad I = S$$

$$(5) \quad S = g(\pi)Y \text{ avec } g'(\pi) > 0$$

$$(6) \quad I = f(\sigma, \pi)Y \text{ avec } f'_\sigma > 0 \text{ et } f'_\pi > 0 \text{ et, de plus } g'(\pi) > f'_\pi(\sigma, \pi)$$

$$(7) \quad \frac{\dot{\sigma}}{\sigma} = \frac{\dot{Y}}{Y} - \frac{\dot{K}}{K}$$

$$(8) \quad \pi = \theta(\sigma) \text{ avec } \theta'(\sigma) > 0$$

On en tire un système différentiel en ε et σ .

Système différentiel en (ε, σ)

Pour le taux d'emploi ε , on a par (2) : $\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{Y}}{Y} - n$, puis par (3) : $\frac{\dot{\varepsilon}}{\varepsilon} = h(\pi, \varepsilon) - n$

Pour le taux d'utilisation des capacités de production σ , on a par (7) : $\frac{\dot{\sigma}}{\sigma} = \frac{\dot{Y}}{Y} - \frac{\dot{K}}{K}$, puis par

$$(1) \text{ et } (3) : \frac{\dot{\sigma}}{\sigma} = h(\pi, \varepsilon) - \frac{I}{K} + \delta = h(\pi, \varepsilon) - \frac{I}{Y} \frac{Y}{K} + \delta, \text{ enfin par } (6) : \frac{\dot{\sigma}}{\sigma} = h(\pi, \varepsilon) - f(\sigma, \pi)\sigma + \delta$$

En se servant de (8), on obtient le système différentiel :

$$(SDS) \quad \begin{cases} \dot{\varepsilon} = \varepsilon [h(\theta(\sigma), \varepsilon) - n] \\ \dot{\sigma} = \sigma [h(\theta(\sigma), \varepsilon) - f(\sigma, \theta(\sigma))\sigma + \delta] \end{cases}$$

Le travail avec Maple peut démarrer ...

```
> restart;
Feps:=(x,y,n)->x*(h(theta(y),x)-n);
Fsig:=(x,y,delta)->y*(h(theta(y),x)-f(theta(y),y)*y+delta);
      Feps := (x, y, n) → x (h(θ(y), x) - n)
      Fsig := (x, y, δ) → y (h(θ(y), x) - f(θ(y), y) y + δ) (2.3.1)
```

A ce niveau de généralité, on pressent que la résolution du système va présenter des difficultés. Il faudra tôt ou tard se contenter de simulations en explicitant les fonctions f , h et θ , sans compter les paramètres.

Equilibres

Tentons la recherche des équilibres avec `solve`.

```
> solve({Feps(epsilon,sigma,n),Fsig(epsilon,sigma,delta)},
[epsilon,sigma]);
Warning, solutions may have been lost
[[ε=0, σ=0], [ε=RootOf(-h(θ(0),_Z)+n), σ=0], [ε=0, σ=RootOf(f(θ(_Z),
_Z)_Z-h(θ(_Z),0)-δ)]] (2.4.1)
```

Maple ne détecte que les équilibres triviaux (cas d'un taux d'emploi nul ou d'un taux d'utilisation du capital nul). Mais il signale que d'autres solutions peuvent exister. Cette prudence est louable puisqu'il existe bel et bien une autre solution.

Si on écarte les cas triviaux, il s'agit de chercher les solutions du système :

$$\begin{cases} h(\theta(\sigma), \varepsilon) = n \\ f(\sigma, \theta(\sigma))\sigma = n + \delta \end{cases}$$

Considérons la seconde équation. Posons $F(\sigma) = f(\sigma, \theta(\sigma))\sigma$. Sa dérivée première vaut $F'(\sigma) = f(\sigma, \theta(\sigma)) + \sigma(f'_\sigma + f'_\pi \theta')$. Comme $f(\sigma, \theta(\sigma)) > 0$, $\sigma > 0$, $f'_\sigma > 0$, $f'_\pi > 0$, $\theta' > 0$, on a $F'(\sigma) > 0$ et la fonction F est monotone croissante à partir de $F(0) = 0 \times f(0, \theta(0)) = 0$. L'équation $F(\sigma) = n + \delta$ admettra une solution unique à condition que le réel $n + \delta$ appartienne à l'ensemble d'arrivée de F . Notons σ^e cette solution.

Considérons alors l'équation $h(\theta(\sigma^e), \varepsilon) = n$. Posons $G(\varepsilon) = h(\theta(\sigma^e), \varepsilon)$. Comme $G'(\varepsilon) = h'(\varepsilon) < 0$, la fonction G est monotone décroissante. Puisque $0 \leq \varepsilon \leq 1$, l'équation $G(\varepsilon) = 0$ admet une solution unique si la fonction G vérifie $G(0) = h(\theta(\sigma^e), 0) \geq n$ et $G(1) = h(\theta(\sigma^e), 1) \leq n$. Notons ε^e cette solution.

Sous les hypothèses citées pour les fonctions F et G , le système différentiel (SDS) admet un équilibre non trivial unique $E = (\varepsilon^e, \sigma^e)$, dont on cherche maintenant les propriétés locales.

Propriétés locales de l'équilibre non trivial

On commence par construire la matrice jacobienne du système (SDS).

```
> Jac:=Matrix(2,2,[[diff(Feps(epsilon,sigma,n),epsilon),diff
(Feps(epsilon,sigma,n),sigma)],[diff(Fsig(epsilon,sigma,
delta),epsilon),diff(Fsig(epsilon,sigma,delta),sigma)]]);
Jac := [[ [h(θ(σ), ε) - n + ε D2(h)(θ(σ), ε), ε D1(h)(θ(σ), ε) ( d/dσ θ(σ) ) ]], (2.5.1)
```

$$\left[\begin{array}{l} \sigma D_2(h)(\theta(\sigma), \varepsilon), h(\theta(\sigma), \varepsilon) - f(\theta(\sigma), \sigma) \sigma + \delta + \sigma \left(D_1(h)(\theta(\sigma), \varepsilon) \left(\frac{d}{d\sigma} \theta(\sigma) \right) - \left(D_1(f)(\theta(\sigma), \sigma) \left(\frac{d}{d\sigma} \theta(\sigma) \right) + D_2(f)(\theta(\sigma), \sigma) \right) \sigma - f(\theta(\sigma), \sigma) \right) \right] \end{array} \right]$$

On cherche l'expression prise par la jacobienne au point d'équilibre non trivial. En tenant compte des résultats du paragraphe précédent, on obtient :

```
> Jac_eq:=Matrix(2,2,[[epsilon*(D[2](h))(theta(sigma), epsilon),
epsilon],Jac[1,2]],[[Jac[2,1],sigma*((D[1](h))(theta(sigma),
epsilon)*(diff(theta(sigma), sigma))-(D[1](f))(theta(sigma),
sigma)*(diff(theta(sigma), sigma)))+(D[2](f))(theta(sigma),
sigma))*sigma-f(theta(sigma), sigma)]]);
```

$$Jac_eq := \left[\left[\begin{array}{l} \varepsilon D_2(h)(\theta(\sigma), \varepsilon), \varepsilon D_1(h)(\theta(\sigma), \varepsilon) \left(\frac{d}{d\sigma} \theta(\sigma) \right) \right], \right. \end{array} \right] \quad (2.5.2)$$

$$\left[\begin{array}{l} \sigma D_2(h)(\theta(\sigma), \varepsilon), \sigma \left(D_1(h)(\theta(\sigma), \varepsilon) \left(\frac{d}{d\sigma} \theta(\sigma) \right) - \left(D_1(f)(\theta(\sigma), \sigma) \left(\frac{d}{d\sigma} \theta(\sigma) \right) + D_2(f)(\theta(\sigma), \sigma) \right) \sigma - f(\theta(\sigma), \sigma) \right) \right] \end{array} \right]$$

Maple retourne facilement le déterminant et la trace de **Jac_eq**.

```
> with(LinearAlgebra):
detJ:=simplify(Determinant(Jac_eq));
detJ:=collect(detJ,[(D[2](h))(theta(sigma), epsilon),epsilon,
sigma]);
traJ:=simplify(Trace(Jac_eq));
```

$$detJ := -\varepsilon D_2(h)(\theta(\sigma), \varepsilon) \sigma^2 D_1(f)(\theta(\sigma), \sigma) D(\theta)(\sigma) - \varepsilon D_2(h)(\theta(\sigma), \varepsilon) \sigma^2 D_2(f)(\theta(\sigma), \sigma) - \varepsilon D_2(h)(\theta(\sigma), \varepsilon) \sigma f(\theta(\sigma), \sigma)$$

$$detJ := \left(\left(-D_1(f)(\theta(\sigma), \sigma) D(\theta)(\sigma) - D_2(f)(\theta(\sigma), \sigma) \right) \sigma^2 - f(\theta(\sigma), \sigma) \sigma \right) \varepsilon D_2(h)(\theta(\sigma), \varepsilon)$$

$$traJ := \varepsilon D_2(h)(\theta(\sigma), \varepsilon) + \sigma D_1(h)(\theta(\sigma), \varepsilon) D(\theta)(\sigma) - \sigma^2 D_1(f)(\theta(\sigma), \sigma) D(\theta)(\sigma) - \sigma^2 D_2(f)(\theta(\sigma), \sigma) - f(\theta(\sigma), \sigma) \sigma \quad (2.5.3)$$

En écriture "ordinaire", le déterminant **detJ** vaut : $\varepsilon \sigma h'_\varepsilon \left(-f - \sigma \left(f'_\pi \theta' + f'_\sigma \right) \right)$ et la trace **traJ** : $\varepsilon h'_\varepsilon + \sigma \left(h'_\pi \theta' - f - \sigma f'_\pi \theta' - \sigma f'_\sigma \right)$. En tenant compte des contraintes sur les fonctions impliquées (pour rappel : $f > 0, f'_\sigma > 0, f'_\pi > 0, \theta' > 0, h'_\pi > 0$ et $h'_\varepsilon < 0$), on voit que le déterminant est positif et que le signe de la trace est indéterminé. Ainsi, les deux valeurs propres sont de même signe, mais ce signe est inconnu. Il pourrait sembler que cette configuration est propice à la recherche d'une bifurcation mais, à raison, ce n'est pas la voie choisie par Skott car aucun paramètre ne se dégage de l'expression de la trace pour en fixer le signe. L'auteur va chercher à montrer que (SDS) peut générer un cycle-limite en satisfaisant les conditions du théorème de Poincaré-Bendixon (ce qui l'oblige à ajouter de nouvelles hypothèses économiques). Nous allons nous démarquer de l'argumentation développée dans son article pour entrer dans des

simulation dont l'objectif est de mettre au point des fonctions de comportement capables de générer un cycle limite. D'ailleurs, ce choix est loin d'être un exercice futile : le recours à Maple dans l'étude du système (SDS) montre à qui veut bien prendre la peine de lire avec attention l'article de Skott que bien des aspects du problème ont été (volontairement?) esquivés par Skott lui même.

Simulations et mise en évidence d'un cycle limite

Il est temps de poser les équations différentielles en Maple.

```
> eqde:=diff(epsilon(t),t)=Feps(epsilon(t),sigma(t),n);
eqds:=diff(sigma(t),t)=Fsig(epsilon(t),sigma(t),delta);
```

$$eqde := \frac{d}{dt} \varepsilon(t) = \varepsilon(t) (h(\theta(\sigma(t)), \varepsilon(t)) - n)$$

$$eqds := \frac{d}{dt} \sigma(t) = \sigma(t) (h(\theta(\sigma(t)), \varepsilon(t)) - f(\theta(\sigma(t)), \sigma(t)) \sigma(t) + \delta) \quad (2.6.1)$$

Les simulations nécessitent de fixer les paramètres et les fonctions de comportement.

Arbitrairement, le taux de croissance démographique et le taux de dépréciation du capital sont annulés : $n = \delta = 0$. La fonction du taux de croissance espéré par les firmes est volontairement

simple : $h(\pi, \varepsilon) = -\varepsilon - \frac{3}{2\pi} + 1$. La fonction d'investissement est définie par :

$f(\sigma, \pi) = -a + \frac{1-b}{\sigma} - \frac{3}{2\pi\sigma}$ avec $b < 1$. Enfin, la relation entre la part des profits et le taux

d'utilisation des capacités de production est linéaire : $\pi = \frac{15}{2}\sigma$. Les dérivées de ces fonctions

satisfont les hypothèses du modèle.

```
> n,delta:=0,0:
theta:=x->(15/2)*x:
h:=(x,y)->-y-(3/2)*(1/x)+1:
f:=(x,y)->-a+(1-b)/y-(3/2)/(x*y):
#récupération des équations différentielles
eqdes:=eqde;
eqdss:=simplify(eqds);
```

$$eqdes := \frac{d}{dt} \varepsilon(t) = \varepsilon(t) \left(-\varepsilon(t) - \frac{1}{5\sigma(t)} + 1 \right)$$

$$eqdss := \frac{d}{dt} \sigma(t) = \sigma(t) (-\varepsilon(t) + a\sigma(t) + b) \quad (2.6.2)$$

Les paramètres a et b introduits dans la fonction f vont permettre de distinguer les cas de stabilité et d'instabilité de l'équilibre non trivial. On voit qu'ils interviennent dans le second membre de l'équation différentielle **eqdss** et par conséquent sur le positionnement de la droite-isocline $\dot{\sigma} = 0$ dans le plan des phases. Signalons au passage que le choix des fonctions de comportement a été guidé par le souci de faire toutes les simulations avec la même hyperbole-isocline $\dot{\varepsilon} = 0$ et une droite-isocline $\dot{\sigma} = 0$.

Simulation 1

La stabilité de l'équilibre est garantie si la pente de l'isocline $\dot{\sigma} = 0$ est inférieure à la pente de l'isocline $\dot{\varepsilon} = 0$, car alors le déterminant de la jacobienne est positif et sa trace est négative.

```
> #injection des paramètres
a,b:=-0.1,0.7;
#récupération des équations différentielles
```


```

eqdes1:=eqdes;
eqdss1:=eqdss;
#chargement des paquetages
with(plots):
with(DEtools):
#construction des isoclines
g_isoe:=plot(solve(subs({epsilon(t)=epsilon,sigma(t)=sigma},
rhs(eqdes1)/epsilon(t)),epsilon),sigma=0..2,epsilon=0..1):
g_isos:=plot(solve(subs({epsilon(t)=epsilon,sigma(t)=sigma},
rhs(eqdss1)/sigma(t)),epsilon),sigma=0..2,epsilon=0..1):
#construction du portrait de phase avec 2 orbites
g_orbs1:=DEplot({eqdes1,eqdss1},[sigma(t),epsilon(t)],t=0.
.100,sigma=0..2,epsilon=0..1,[[sigma(0)=0.5,epsilon(0)=0.12],
[sigma(0)=1.6,epsilon(0)=0.95]],stepsize=0.05,linecolor=
black,thickness=1,scene=[sigma,epsilon]):
#graphique final réunissant le portrait de phase et les
isoclines
display(g_isoe,g_isos,g_orbs1);
a, b := -0.1, 0.7

```

$$eqdes1 := \frac{d}{dt} \varepsilon(t) = \varepsilon(t) \left(-\varepsilon(t) - \frac{1}{5\sigma(t)} + 1 \right)$$

$$eqdss1 := \frac{d}{dt} \sigma(t) = \sigma(t) (-\varepsilon(t) - 0.1\sigma(t) + 0.7)$$



Simulation 2

L'équilibre est instable si la pente de l'isocline $\dot{\sigma} = 0$ est supérieure à la pente de l'isocline $\dot{\varepsilon} = 0$, car alors le déterminant de la jacobienne est positif et sa trace est négative. On procède de la même manière que dans la simulation 1 pour illustrer cette propriété.

```


> a,b:=1,-0.25;
eqdes1:=eqdes;
eqdss1:=eqdss;
with(plots):
with(DEtools):
g_iseo:=plot(solve(subs({epsilon(t)=epsilon,sigma(t)=sigma},
rhs(eqdes1)/epsilon(t)),epsilon),sigma=0..2,epsilon=0..1):
g_isos:=plot(solve(subs({epsilon(t)=epsilon,sigma(t)=sigma},
rhs(eqdss1)/sigma(t)),epsilon),sigma=0..2,epsilon=0..1):
g_orbs1:=DEplot({eqdes1,eqdss1},[sigma(t),epsilon(t)],t=0.
.100,sigma=0..2,epsilon=0..1,[[sigma(0)=1.15,epsilon(0)=
0.95],[sigma(0)=0.9,epsilon(0)=0.5],[sigma(0)=0.8,epsilon(0)=
0.5],[sigma(0)=1.1,epsilon(0)=0.95]],stepsize=0.05,linecolor=
black,thickness=1,scene=[sigma,epsilon]):
display(g_iseo,g_isos,g_orbs1);

```

$a, b := 1, -0.25$

$$eqdes1 := \frac{d}{dt} \varepsilon(t) = \varepsilon(t) \left(-\varepsilon(t) - \frac{1}{5\sigma(t)} + 1 \right)$$

$$eqdss1 := \frac{d}{dt} \sigma(t) = \sigma(t) (-\varepsilon(t) + \sigma(t) - 0.25)$$



Ces deux premières simulations montrent que deux cas seulement sont possibles si l'isocline $\dot{\sigma} = 0$ est une droite : l'équilibre non trivial est stable ou instable suivant la pente de cette droite. Il faut donc abandonner les formulations données au départ pour obtenir un cycle-limite. Une chose est sûre : on doit conserver l'idée que la pente de l'isocline $\dot{\sigma} = 0$ doit être supérieure à celle

de l'isocline $\dot{\epsilon} = 0$ pour que l'équilibre soit répulsif. Mais il faut aussi que cet équilibre soit attracteur à distance! Ce challenge est bien connu en mathématiques appliquées. Si l'isocline $\dot{\sigma} = 0$ est une sigmoïde appropriée, l'équilibre reste instable, mais il existe un cycle limite structurellement stable dont il faut prouver l'existence avec le théorème de Poincaré-Bendixon. Autant dire que l'économiste doit se pencher sur deux problèmes : forger des arguments économiques convaincants pour justifier la sigmoïde; montrer que le théorème de Poincaré-Bendixon s'applique. Sur le premier point, on renvoie à l'article de Skott (le lecteur ne sera pas nécessairement séduit par ses arguments!). Sur le second, on se contente ici d'illustrer le phénomène d'apparition d'un cycle limite quand l'isocline $\dot{\sigma} = 0$ est sigmoïdale (via la construction d'un polynôme de degré 3 soigneusement situé dans le plan des phases) et l'isocline $\dot{\epsilon} = 0$ est hyperbolique (elle aussi, stratégiquement située). La figure terminale donne le champ directionnel, des orbites "fuyant" l'équilibre instable pour converger vers un cycle limite, des orbites attirées par l'équilibre et convergeant vers le même cycle limite. L'ensemble compact invariant nécessaire pour faire fonctionner le théorème de Poincaré-Bendixon a été bleuté.


```
> #construction de l'isocline sigmoïdale
Pol:=x->a1*x^3+a2*x^2+a3*x+a4:#choix d'une fonction
polynômiale de degré 3
coef:=solve({Pol(0.2)=1,Pol(1)=0.4,Pol(2)=0.5,Pol(4)=0},{a1,
a2,a3,a4}):#fixation des paramètres pour donner une allure
convenable à l'isocline
assign(coef):
Pol(x):
FPol:=unapply(Pol(x),x):
g1:=plot(FPol(sigma),sigma=0..4,epsilon=0..1):#graphe de
l'isocline sigma'=0

#construction de l'isocline hyperbolique
H:=(s,e)->-0.5/(0.625*s+0.1)-0.4*e+0.61:
solve(H(s,e)=0,e):
H1:=unapply(%,s):
g2:=plot(H1(sigma),sigma=0..4,epsilon=0..1):#graphe de
l'isocline epsilon'=0

#construction des orbites et du champ directionnel dans le
plan des phases
eqd1:=diff(epsilon(t),t)=epsilon(t)*(H(sigma(t),epsilon(t)))
:#équation différentielle en epsilon
eqd2:=diff(sigma(t),t)=sigma(t)*(Pol(sigma(t))-epsilon(t))
:#équation différentielle en sigma
g_orbs:=DEplot({eqd1,eqd2},{epsilon(t),sigma(t)},t=0..150,
sigma=0..4,epsilon=0..1,[[sigma(0)=0.5,epsilon(0)=0.1],[sigma
(0)=1.5,epsilon(0)=0.4],[sigma(0)=2,epsilon(0)=0.5],[sigma(0)
=3,epsilon(0)=0.9]],stepsize=0.05,linecolor=black,thickness=
1,scene=[sigma,epsilon]):#résolution numérique avec 4
conditions initiales

:#construction d'un ensemble invariant
g_inv:=polygonplot([[0,0],[0,1],[3.68,1],[4,0]], color =HUE
(0.5), transparency = .8):

#graphique terminal
display({g1,g2,g_orbs,g_inv},labels=[``,``]);#graphique final
```



▼ Modèle de van der Ploeg (1987)

Van der Ploeg publie en 1987 un article important et très profond dans lequel il endogénéise la dynamique de la productivité du travail dans un modèle de croissance cyclique. Il conserve donc la thématique du conflit entre les salariés et les capitalistes sur la répartition du revenu national (comme chez Goodwin) mais aussi sur le partage des gains de productivité du travail en lui ajoutant un mécanisme de changement technique induit par l'état des rapports sociaux de répartition, inspiré des intuitions de Shah et Desai (1981). Il explore les propriétés d'un système dynamique tridimensionnel non linéaire dont les variables d'état sont le taux d'emploi, la part des salaires dans le revenu et le capital-output ratio. Une bifurcation de Hopf est économiquement plausible et fait apparaître un cycle limite.

▼ Présentation générale

On adoptera les notations suivantes :

1. Y , K , I , S , Π , L , et N désignent respectivement le revenu, le capital, l'investissement, l'épargne, le profit, la population active occupée et la population active totale.

2. le taux de salaires est w .

3. la part des salaires dans le revenu est $\omega = \frac{wL}{Y}$.

4. la productivité moyenne du travail est $a = \frac{Y}{L}$

5. le capital-output ratio est $\sigma = \frac{K}{Y}$

Comme dans la plupart des modèles post-goodwiniens, le contenu économique du modèle embrasse les sphères des revenus, des échanges et de la production sur lesquelles se greffent les conflits sociaux de répartition.

Dans la sphère des revenus, le revenu national est partagé entre les salaires, wL , et les profits, Π : $Y = wL + \Pi$. Tous les agents épargnent une fraction de leurs revenus mais, à la manière cambridgienne, le taux d'épargne des capitalistes s_1 est plus élevé que celui des salariés s_2 .

L'épargne nationale vaut $S = s_1(Y - wL) + s_2wL$ avec $0 \leq s_2 < s_1 \leq 1$.

Dans la sphère des échanges, l'équilibre macroéconomique équivaut à l'égalité de l'épargne et de l'investissement. On pose que toute l'épargne est investie : $I = S$. Dans l'hypothèse peu crédible mais commode (parce qu'elle simplifie les calculs ultérieurs) d'un taux de dépréciation nul du capital productif, $I = \dot{K}$. L'équilibre se réécrit : $\dot{K} = s_1(Y - wL) + s_2wL$. S'en déduit le taux

$$\text{d'accumulation du capital : } \frac{\dot{K}}{K} = \frac{s_1(Y - wL) + s_2wL}{\sigma Y} = \frac{s_1(1 - \omega) + s_2\omega}{\sigma}$$
$$= \frac{s_1 - (s_1 - s_2)\omega}{\sigma}$$

Le rythme de croissance du capital dépend négativement de la part des salaires dans le revenu national (donc, positivement de la part des profits) et négativement du capital-output ratio.

Dans la sphère productive, l'emploi répond à des considérations techniques et le choix de la combinaison productive à des stratégies microéconomiques.

L'évolution du taux d'emploi ε est donnée par la différence entre les variations relatives de la population active occupée et de la population active totale, supposée croître au taux

$$\text{démographique constant } n : \frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{L}}{L} - \frac{\dot{N}}{N} = \frac{\dot{L}}{L} - n.$$

La combinaison technologique choisie par les capitalistes découle d'une stratégie qui tient compte des variations de la répartition. L'argumentation de van der Ploeg, inspirée des travaux antérieurs de Shah et Desai, est non standard.

Les facteurs de production capital et travail sont substituables le long d'une frontière de production "à la Kaldor". Chez Kaldor, les gains de productivité du travail sont liés à la croissance de l'intensité capitaliste par l'invocation du principe des rendements d'échelle dynamiques. On a directement $\frac{\dot{a}}{a} = \chi\left(\frac{\overline{K/L}}{K/L}\right)$ avec $\chi(0) \geq 0$, $\chi'\left(\frac{\overline{K/L}}{K/L}\right) > 0$ et $\chi''\left(\frac{\overline{K/L}}{K/L}\right) \leq 0$

parce que les gains de productivité sont croissants avec le rythme d'alourdissement du processus de production avec un phénomène de décroissance marginale. Van der Ploeg déforme

le concept kaldorien d'intensité capitaliste en le mesurant par $\sigma = \frac{K}{Y}$, donc par le capital-output ratio plutôt que par le rapport $\frac{K}{L}$, de sorte qu'il retient la frontière technologique

$$\frac{\dot{a}}{a} = \chi\left(\frac{\dot{\sigma}}{\sigma}\right) \text{ avec } \chi(0) \geq 0, \chi'\left(\frac{\dot{\sigma}}{\sigma}\right) > 0 \text{ et } \chi''\left(\frac{\dot{\sigma}}{\sigma}\right) \leq 0.$$

Le choix de la technique résulte d'un comportement rationnel de maximisation de la croissance du taux de profit sous la contrainte technologique. Notons $r = \frac{\Pi}{K}$ le taux de profit. On a :

$$r = \frac{\Pi}{Y} \frac{Y}{K} = \frac{1}{\sigma} \left(1 - \frac{wL}{Y} \right), \text{ soit, en variations relatives : } \frac{\dot{r}}{r} = -\frac{\dot{\sigma}}{\sigma} - \frac{\frac{w}{a}}{1 - \frac{w}{a}} \frac{\dot{\frac{w}{a}}}{\frac{w}{a}} = -\frac{\dot{\sigma}}{\sigma} - \frac{\frac{\dot{w}}{w}}{1 - \frac{w}{a}} \frac{a}{w}$$

En supposant que la décision est prise en tenant les salaires pour une donnée fixe : $\frac{\dot{a}}{a} = -\frac{\dot{w}}{w}$. Ainsi, l'objectif est de choisir le couple $\left(\frac{\dot{a}}{a}, \frac{\dot{\sigma}}{\sigma} \right)$ qui maximise

$$\frac{\dot{r}}{r} = -\frac{\dot{\sigma}}{\sigma} + \frac{\omega}{1 - \omega} \frac{\dot{a}}{a}, \text{ sous la contrainte technologique.}$$

Le programme max $\frac{\dot{r}}{r} = -\frac{\dot{\sigma}}{\sigma} + \frac{\omega}{1 - \omega} \frac{\dot{a}}{a}$ s.c. $\frac{\dot{a}}{a} = \chi \left(\frac{\dot{\sigma}}{\sigma} \right)$ équivaut à

$$\begin{aligned} \max \frac{\dot{r}}{r} &= -\frac{\dot{\sigma}}{\sigma} + \frac{\omega}{1 - \omega} \chi \left(\frac{\dot{\sigma}}{\sigma} \right). \text{ La condition du premier ordre amène : } -1 + \frac{\omega}{1 - \omega} \chi' \left(\frac{\dot{\sigma}}{\sigma} \right) \\ &= 0, \text{ dont on tire } \frac{\dot{\sigma}}{\sigma} = \chi'^{-1} \left(\frac{1 - \omega}{\omega} \right) \text{ qui correspond à un maximum global puisque } \chi'' \left(\frac{\dot{\sigma}}{\sigma} \right) \\ &\leq 0. \text{ Par la suite, on posera : } \frac{\dot{\sigma}}{\sigma} = \psi(\omega) \text{ avec } \psi \text{ tel que } \psi'(\omega) = -\frac{1}{\omega^2} \frac{1}{\chi''} > 0. \end{aligned}$$

Les rapports sociaux de répartition concernent la dynamique des salaires. Leurs variations s'expliquent par un effet Phillips-Goodwin usuel (la hausse du taux d'emploi se traduit par des gains salariaux) et par un terme de partage des gains de productivité du travail (par indexation automatique d'intensité constante μ). L'addition des deux effets donne : $\frac{\dot{w}}{w} = f(\varepsilon) + \mu \frac{\dot{a}}{a}$ où $f'(\varepsilon) > 0$ et $\mu \geq 0$.

Forme structurelle et système différentiel en $(\sigma, \omega, \varepsilon)$

La forme structurelle du modèle de van der Ploeg comprend 8 équations :

- (1) $\frac{\dot{K}}{K} = \frac{s_1(1 - \omega) + s_2 \omega}{\sigma} = \frac{\theta(\omega)}{\sigma}$ avec $\theta'(\omega) < 0$ car $0 \leq s_2 < s_1 \leq 1$.
- (2) $\frac{\dot{\sigma}}{\sigma} = \psi(\omega)$ avec $\psi'(\omega) = -\frac{1}{\omega^2} \frac{1}{\chi''} > 0$.
- (3) $\frac{\dot{w}}{w} = f(\varepsilon) + \mu \frac{\dot{a}}{a}$ avec $f'(\varepsilon) > 0$ et $\mu \geq 0$
- (4) $\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{L}}{L} - n$

$$(5) \frac{\dot{a}}{a} = \frac{\dot{Y}}{Y} - \frac{\dot{L}}{L}$$

$$(6) \frac{\dot{\omega}}{\omega} = \frac{\dot{w}}{w} + \frac{\dot{L}}{L} - \frac{\dot{Y}}{Y}$$

$$(7) \frac{\dot{\sigma}}{\sigma} = \frac{\dot{K}}{K} - \frac{\dot{Y}}{Y}$$

$$(8) \frac{\dot{a}}{a} = \chi\left(\frac{\dot{\sigma}}{\sigma}\right) \text{ avec } \chi(0) \geq 0, \chi'\left(\frac{\dot{\sigma}}{\sigma}\right) > 0 \text{ et } \chi''\left(\frac{\dot{\sigma}}{\sigma}\right) \leq 0.$$

On en déduit un système différentiel en σ , ω et ε (l'ordre des variables est celui choisi par van der Ploeg).

L'évolution du capital-output ratio est donnée par (2).

L'évolution de la part des salaires est la différence entre la croissance des salaires et celle de la productivité du travail : $\frac{\dot{\omega}}{\omega} = \frac{\dot{w}}{w} - \frac{\dot{a}}{a}$ (par (5) et (6)). Par (3) : $\frac{\dot{\omega}}{\omega} = f(\varepsilon) + (\mu - 1) \frac{\dot{a}}{a}$.

Puis par (8) : $\frac{\dot{\omega}}{\omega} = f(\varepsilon) + (\mu - 1) \chi\left(\frac{\dot{\sigma}}{\sigma}\right)$. Enfin, par (2) : $\frac{\dot{\omega}}{\omega} = f(\varepsilon) + (\mu - 1)$

$\chi(\psi(\omega))$.

L'évolution du taux d'emploi est donnée par (4) : $\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{L}}{L} - n$. Par (5) : $\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{Y}}{Y} - \frac{\dot{a}}{a} - n$. Puis

par (8) et (2) : $\frac{\dot{\varepsilon}}{\varepsilon} = \frac{\dot{Y}}{Y} - \chi(\psi(\omega)) - n$. Enfin par (7), (2) et (1) : $\frac{\dot{\varepsilon}}{\varepsilon} =$

$\frac{\theta(\omega)}{\sigma} - \psi(\omega) - \chi(\psi(\omega)) - n$.

Le système différentiel à étudier est le suivant :

$$(SDVDP) \begin{cases} \frac{\dot{\sigma}}{\sigma} = \psi(\omega) \\ \frac{\dot{\omega}}{\omega} = f(\varepsilon) + (\mu - 1) \chi(\psi(\omega)) \\ \frac{\dot{\varepsilon}}{\varepsilon} = \frac{\theta(\omega)}{\sigma} - \psi(\omega) - \chi(\psi(\omega)) - n \end{cases}$$

Pour mémoire, rappelons les principales hypothèses sur les fonctions de comportement de ce modèle :

$\psi'(\omega) > 0$: une augmentation de la part des salaires accroît le capital par unité produite

$f'(\varepsilon) > 0$: une augmentation du taux d'emploi amène des augmentations salariales

$\theta'(\omega) < 0$: une augmentation de la part des salaires diminue l'accumulation du capital

$\chi'\left(\frac{\dot{\sigma}}{\sigma}\right) > 0$: l'accumulation du capital par unité d'output accroît les gains de productivité du travail

```
> restart;
F1:=(x,y,z)->x*psi(y);
F2:=(x,y,z)->y*(f(z)+(mu-1)*(chi@psi)(y));
F3:=(x,y,z)->z*((1/x)*theta(y)-psi(y)-(chi@psi)(y)-n);
```

$$\begin{aligned}
F1 &:= (x, y, z) \rightarrow x \psi(y) \\
F2 &:= (x, y, z) \rightarrow y (f(z) + (\mu - 1) \chi @ \psi(y)) \\
F3 &:= (x, y, z) \rightarrow z \left(\frac{\theta(y)}{x} - \psi(y) - \chi @ \psi(y) - n \right)
\end{aligned} \tag{3.2.1}$$

A ce niveau de généralité, il est exclu de chercher une solution explicite, d'autant que ce système différentiel tri-dimensionnel est non linéaire. On cherche donc à cerner qualitativement les propriétés dynamiques de (SDVDP) : détection des équilibres; étude de leur stabilité locale; avec le concours de Maple, tracer des orbites dans l'espace des phases suivant les valeurs prises par les paramètres.

Equilibre

Comme d'habitude, on lance la commande **solve**. Elle se révèle instable dans ce cas de figure. Selon l'état du système informatique au moment où on appuie sur la touche Enter, la réponse est une liste vide ou une solution unique.

```
> Equi:=solve({F1(x,y,z)=0,F2(x,y,z)=0,F3(x,y,z)=0},[x,y,z]);
#cas où la réponse est une liste vide
Equi := [ ]
```

(3.3.1)

```
> Equi:=solve({F1(x,y,z)=0,F2(x,y,z)=0,F3(x,y,z)=0},[x,y,z]);
#cas où la réponse est une solution unique
Equi := [[ x =  $\frac{\theta(\text{RootOf}(\psi(\_Z)))}{\chi(0) + n}$ , y =  $\text{RootOf}(\psi(\_Z))$ , z =  $\text{RootOf}(f(\_Z) + \chi(0) \mu$ 
-  $\chi(0))$  ]]
```

(3.3.2)

La seconde réponse est intéressante. Première bonne chose : Maple a tenu compte du fait que x ne peut s'annuler. Deuxième bonne chose : s'il existe, l'équilibre est unique. Troisième remarque : l'existence d'un équilibre est suspendue au fait que l'équation $\psi(\omega) = 0$ admet une solution unique. Regardons ce point de plus près.

Par construction : $\psi = \chi'^{-1} \circ H$ où $H(x) = \frac{1-x}{x}$. Donc $\psi^{-1} = (\chi'^{-1} \circ H)^{-1} = H^{-1} \circ \chi'$. Mais $H^{-1}(y) = \frac{1}{1+y}$, donc $\psi^{-1} = \frac{1}{1+\chi'}$ et la part des salaires à l'équilibre vérifie $\omega^e = \psi^{-1}(0)$

$$= \frac{1}{1+\chi'(0)} > 0. \text{ On en déduit : } \varepsilon^e = f^{-1}((1-\mu)\chi(0)) \text{ et } \sigma^e = \frac{\theta(\omega^e)}{\chi(0) + n}.$$

En conclusion, (SDVDP) admet un équilibre dynamique unique

$$E = \left(\frac{\theta(\omega^e)}{\chi(0) + n}, f^{-1}((1-\mu)\chi(0)), \frac{1}{1+\chi'(0)} \right).$$

Stabilité de l'équilibre

La stabilité de l'équilibre s'étudie au niveau local par linéarisation de (SDVDP). Première étape : calcul de la jacobienne.

```
> Jac:=Matrix(3,3,[ [diff(F1(sigma,omega,epsilon),sigma),diff(F1(sigma,omega,epsilon),omega),diff(F1(sigma,omega,epsilon),epsilon)], [diff(F2(sigma,omega,epsilon),sigma),diff(F2(sigma,omega,epsilon),omega),diff(F2(sigma,omega,epsilon),epsilon)], [diff(F3(sigma,omega,epsilon),sigma),diff(F3(sigma,omega,epsilon),omega),diff(F3(sigma,omega,epsilon),epsilon)],
```

```
[diff(F3(sigma,omega,epsilon),sigma),diff(F3(sigma,omega,epsilon),omega),diff(F3(sigma,omega,epsilon),epsilon)]];
```

$$Jac := \begin{bmatrix} \psi(\omega), \sigma \left(\frac{d}{d\omega} \psi(\omega) \right), 0 \end{bmatrix}, \quad (3.4.1)$$

$$\begin{bmatrix} 0, f(\varepsilon) + (\mu - 1) \chi(\psi(\omega)) + \omega (\mu - 1) D(\chi)(\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right), \\ \omega \left(\frac{d}{d\varepsilon} f(\varepsilon) \right) \end{bmatrix}$$

$$\begin{bmatrix} -\frac{\varepsilon \theta(\omega)}{\sigma^2}, \varepsilon \left(\frac{\frac{d}{d\omega} \theta(\omega)}{\sigma} - \left(\frac{d}{d\omega} \psi(\omega) \right) - D(\chi)(\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \right), \\ \frac{\theta(\omega)}{\sigma} - \psi(\omega) - \chi(\psi(\omega)) - n \end{bmatrix}$$

Deuxième étape : calcul de la jacobienne à l'équilibre E .

```
> Jac[1,1]:=0;
Jac[2,2]:=omega*(mu-1)*(D(chi))(psi(omega))*(diff(psi(omega),omega));
Jac[3,3]:=0;
Jac_eq:=Jac;
```

$$Jac_eq := \begin{bmatrix} \psi(\omega), \sigma \left(\frac{d}{d\omega} \psi(\omega) \right), 0 \end{bmatrix}, \quad (3.4.2)$$

$$\begin{bmatrix} 0, \omega (\mu - 1) D(\chi)(\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right), \omega \left(\frac{d}{d\varepsilon} f(\varepsilon) \right) \end{bmatrix}$$

$$\begin{bmatrix} -\frac{\varepsilon \theta(\omega)}{\sigma^2}, \varepsilon \left(\frac{\frac{d}{d\omega} \theta(\omega)}{\sigma} - \left(\frac{d}{d\omega} \psi(\omega) \right) - D(\chi)(\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \right), 0 \end{bmatrix}$$

Les propriétés de l'équilibre sont données par les valeurs propres. Le polynôme caractéristique est : $r^3 - tr(Jac_eq) r^2 + A_2 r - det(Jac_eq)$ où A_2 est la somme de tous les déterminants des mineurs principaux.

```
> with(LinearAlgebra):
A1:=-Trace(Jac_eq);
A2:=-Jac_eq[2,3]*Jac_eq[3,2];
```

A3:=-Determinant(Jac_eq);
A4:=A1*A2-A3;

$$\begin{aligned}
 A1 &:= -\omega (\mu - 1) D(\chi) (\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \\
 A2 &:= -\omega \left(\frac{d}{d\varepsilon} f(\varepsilon) \right) \varepsilon \left(\frac{\frac{d}{d\omega} \theta(\omega)}{\sigma} - \left(\frac{d}{d\omega} \psi(\omega) \right) \right. \\
 &\quad \left. - D(\chi) (\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \right) \\
 A3 &:= \frac{\varepsilon \theta(\omega) \left(\frac{d}{d\omega} \psi(\omega) \right) \omega \left(\frac{d}{d\varepsilon} f(\varepsilon) \right)}{\sigma} \\
 A4 &:= \omega^2 (\mu - 1) D(\chi) (\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \left(\frac{d}{d\varepsilon} f(\varepsilon) \right) \varepsilon \left(\frac{\frac{d}{d\omega} \theta(\omega)}{\sigma} \right. \\
 &\quad \left. - \left(\frac{d}{d\omega} \psi(\omega) \right) - D(\chi) (\psi(\omega)) \left(\frac{d}{d\omega} \psi(\omega) \right) \right) \\
 &\quad - \frac{\varepsilon \theta(\omega) \left(\frac{d}{d\omega} \psi(\omega) \right) \omega \left(\frac{d}{d\varepsilon} f(\varepsilon) \right)}{\sigma}
 \end{aligned} \tag{3.4.3}$$

D'après le théorème de Routh-Hurwitz, les 3 valeurs propres sont négatives si le critère suivant est satisfait :

$$A3 > 0; A2 > 0; A4 = A1 A2 - A3 > 0$$

Comme $f'(\varepsilon) > 0$, $\theta'(\omega) = s_2 - s_1 < 0$, $\psi'(\omega) > 0$ et $\chi' > 0$, les expressions A3 et A2 sont certainement positives. En revanche, le signe de $A4 = A1 A2 - A3$ est ambigu. De deux choses l'une : soit un des paramètres du modèle a une valeur qui garantit la positivité de A4, et donc les valeurs propres sont négatives et l'équilibre est localement stable; soit il rend A4 négative et l'équilibre est instable. Le candidat naturel est le taux d'indexation des salaires aux gains de productivité μ , mais - et c'est le choix de van der Ploeg - on peut alternativement choisir $\psi'(\omega^e)$. Quoiqu'il en soit, si on se place dans l'optique du théorème de Hopf, on pressent que ce paramètre va jouer un rôle crucial en tant que valeur de bifurcation, faisant passer l'équilibre du statut de point attracteur à celui de répulseur avec apparition d'un cycle-limite. Zhang (1988) démontrera que c'est bien le cas : le modèle de van der Ploeg admet une bifurcation de Hopf super-critique et le cycle-limite est un attracteur. C'est cette propriété qu'on va maintenant illustrer à l'aide de simulations.

Simulations

Les simulations sont basées sur des fonctions les plus simples possibles compatibles avec les hypothèses du modèle. La fonction d'épargne θ et la fonction de Phillips f sont évidemment affines. La frontière de production est logarithmique avec une ordonnée positive à l'origine. La fonction de choix technologique est rationnelle (hyperbolique). Avant d'illustrer l'apparition d'un cycle-limite dans le modèle (SDVDP), on commence par présenter le résultat majeur du modèle


de Shah et Desai, contre lequel van der Ploeg a réagi.

Simulation du modèle de Shah et Desai

Le modèle de Shah et Desai correspond au cas où les salaires ne sont pas indexés sur les gains de productivité et où les salariés n'épargnent rien, de sorte que tout le profit devient l'épargne qui est totalement réinvestie. Dans le modèle (*SDVDP*), ce cas correspond aux valeurs des paramètres : $\mu = 0$, $s_1 = 1$ et $s_2 = 0$, d'où $\theta(\omega) = 1 - \omega$. Nous sommes placés dans les conditions du modèle de Goodwin. La simulation proposée ci dessous confirme la conclusion des auteurs : il y a convergence cyclique amortie vers l'équilibre. On a ainsi perdu la propriété de croissance cyclique!

```
> f:=x->4*x/3-1:
chi:=x->log(1+10*x)+0.1:
psi:=x->(1.1*x-0.1)/(1-x):
theta:=x->1-x:
n:=0.01:
mu:=0:
sysdifs:=diff(sigma(t),t)=F1(sigma(t),omega(t),epsilon(t))
),diff(omega(t),t)=F2(sigma(t),omega(t),epsilon(t)),diff
(epsilon(t),t)=F3(sigma(t),omega(t),epsilon(t));
var:=sigma(t),omega(t),epsilon(t);
with(DEtools):
DEplot3d({sysdifs},{var},t=0..25,[[sigma(0)=3,omega(0)=
0.7,epsilon(0)=0.75],[sigma(0)=3,omega(0)=0.5,epsilon(0)=
0.9],[sigma(0)=3,omega(0)=0.5,epsilon(0)=0.6]],linecolor=
black,labels=[sigma,omega,epsilon],stepsize=0.01,
orientation=[64,84,4],omega=0..1,epsilon=0..1.2,sigma=3.
.5,axes=normal);
```

$$\text{sysdifs} := \frac{d}{dt} \sigma(t) = \frac{\sigma(t) (1.1 \omega(t) - 0.1)}{1 - \omega(t)}, \quad \frac{d}{dt} \omega(t) = \omega(t) \left(\frac{4}{3} \epsilon(t) - 1.1 \right. \\ \left. - \ln \left(1 + \frac{10 (1.1 \omega(t) - 0.1)}{1 - \omega(t)} \right) \right), \quad \frac{d}{dt} \epsilon(t) = \epsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} \right. \\ \left. - \frac{1.1 \omega(t) - 0.1}{1 - \omega(t)} - \ln \left(1 + \frac{10 (1.1 \omega(t) - 0.1)}{1 - \omega(t)} \right) - 0.11 \right) \\ \text{var} := \sigma(t), \omega(t), \epsilon(t)$$



On constate que les orbites convergent toutes vers un point d'équilibre dans l'espace des phases. Shah et Desai interprètent ce résultat par le fait que, comparativement au modèle de Goodwin, les capitalistes disposent d'une arme supplémentaire pour maîtriser les cycles. Ils peuvent modifier la combinaison productive en faveur du capital et au détriment du travail dès que la part des salaires s'accroît trop.

Van der Ploeg rejette ce mécanisme d'amortissement automatique des cycles par contrôle des techniques de production. Certes, les capitalistes peuvent imposer le changement technique, mais les salariés peuvent contrarier les effets de l'accroissement des gains de productivité du travail sur la part des profits en imposant, de leur côté, le principe d'indexation des salaires sur ces mêmes gains. Cette tension sur les conséquences économiques du progrès technique peut même renforcer les mécanismes de croissance cyclique perpétuelle.

▼ Simulations du modèle de van der Ploeg

Comme annoncé en introduction de la section, les simulations de (*SDVDP*) font appel à des fonctions de comportement simple. On fixera arbitrairement le taux de croissance démographique à 0,01. Pour simplifier encore plus la question, on retiendra l'hypothèse "classique" d'un taux d'épargne nul des salariés et un taux d'épargne de 100% pour les capitalistes.

```
> chi:=x->log(a*x+b):
psi:=x->((a+b)*x-b)/(a*(1-x)):
theta:=x->1-x:
f:=x->c*x-d:
n:=0.01:
```

```

mu:='mu':#désassignation de mu (qui avait été fixé à 0
précédemment)
sdvdp:=diff(sigma(t),t)=F1(sigma(t),omega(t),epsilon(t)
),diff(omega(t),t)=F2(sigma(t),omega(t),epsilon(t)),diff
(epsilon(t),t)=F3(sigma(t),omega(t),epsilon(t));

```

$$\begin{aligned}
sdvdp := \frac{d}{dt} \sigma(t) &= \frac{\sigma(t) \left((a+b) \omega(t) - b \right)}{a (1 - \omega(t))}, \quad \frac{d}{dt} \omega(t) = \omega(t) \left(c \varepsilon(t) - d + (\mu \right. \\
&- 1) \ln \left(\frac{(a+b) \omega(t) - b}{1 - \omega(t)} + b \right) \Bigg), \quad \frac{d}{dt} \varepsilon(t) = \varepsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} \right. \\
&\left. - \frac{(a+b) \omega(t) - b}{a (1 - \omega(t))} - \ln \left(\frac{(a+b) \omega(t) - b}{1 - \omega(t)} + b \right) - 0.01 \right)
\end{aligned} \quad (3.5.2.1)$$

Deux options sont possibles pour faire apparaître un cycle-limite : soit μ est le paramètre de bifurcation; soit on prend le rapport $\frac{b}{a}$.

Le paramètre de bifurcation est μ

Intéressons nous aux extrêmes $\mu = 0$ et $\mu = 1$.

Quand l'indexation est nulle, on doit logiquement retrouver la conclusion de Shah et Desai. Cette supposition est confirmée par une représentation graphique d'orbites dans l'espace des phases.

```


> #assignation des paramètres a, b, c et d
a,b,c,d:=0.6,1.2,1.3,1;
#assignation de mu
mu:=0;
#système différentiel
sdvdp1:=sdvdp;
#représentation graphique
DEplot3d({sdvdp1},[var],t=0..25,[[sigma(0)=1.8,omega(0)=
0.7,epsilon(0)=0.91],[sigma(0)=1.7,omega(0)=0.6,epsilon
(0)=0.9],[sigma(0)=1.8,omega(0)=0.6,epsilon(0)=0.89]],
stepsize=0.01,linecolor=[black,blue,red],orientation=
[48,85,-2],labels=[sigma,omega,epsilon],axes=normal);

```

$$a, b, c, d := 0.6, 1.2, 1.3, 1$$

$$\mu := 0$$


$$\begin{aligned}
sdvdp1 := \frac{d}{dt} \sigma(t) &= \frac{1.666666667 \sigma(t) (1.8 \omega(t) - 1.2)}{1 - \omega(t)}, \quad \frac{d}{dt} \omega(t) \\
&= \omega(t) \left(1.3 \varepsilon(t) - 1 - \ln \left(\frac{1.8 \omega(t) - 1.2}{1 - \omega(t)} + 1.2 \right) \right), \quad \frac{d}{dt} \varepsilon(t) \\
&= \varepsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} - \frac{1.666666667 (1.8 \omega(t) - 1.2)}{1 - \omega(t)} - \ln \left(\frac{1.8 \omega(t) - 1.2}{1 - \omega(t)} \right. \right. \\
&\left. \left. + 1.2 \right) - 0.01 \right)
\end{aligned}$$



Quand l'indexation est totale, l'équilibre est instable et toutes les orbites convergent vers un cycle-limite.

```
> #réassignation de mu
mu:=1;
#système différentiel
sdvdp2:=sdvdp;
#représentation graphique
DEplot3d({sdvdp2},[var],t=0..25,[[sigma(0)=1.8,omega(0)=
0.7,epsilon(0)=0.78],[sigma(0)=5,omega(0)=0.6,epsilon(0)
=0.74]],stepsize=0.01,linecolor=[black,red],orientation=
[90,87,-6],labels=[sigma,omega,epsilon],axes=normal);
```

$$\begin{aligned} \mu &:= 1 \\ \text{sdvdp2} &:= \frac{d}{dt} \sigma(t) = \frac{1.666666667 \sigma(t) (1.8 \omega(t) - 1.2)}{1 - \omega(t)}, \frac{d}{dt} \omega(t) \\ &= \omega(t) (1.3 \epsilon(t) - 1), \frac{d}{dt} \epsilon(t) = \epsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} \right. \\ &\quad \left. - \frac{1.666666667 (1.8 \omega(t) - 1.2)}{1 - \omega(t)} - \ln \left(\frac{1.8 \omega(t) - 1.2}{1 - \omega(t)} + 1.2 \right) - 0.01 \right) \end{aligned}$$



Le paramètre d'indexation des salaires sur les gains de productivité du travail admet bien une valeur de bifurcation comprise entre 0 et 1. Au delà de cette valeur, le modèle de croissance cyclique devient structurellement stable car toutes les orbites convergent vers un cycle-limite attracteur.

▼ Le paramètre de bifurcation est $\frac{b}{a}$

Supposons maintenant, comme le fait van der Ploeg, que le paramètre d'indexation est fixé une fois pour toute, par exemple à $\mu = 0,9$. On choisit de faire varier le rapport $\frac{b}{a}$ pour étudier le comportement qualitatif de $(SDVDP)$, les paramètres c et d étant fixés à 1,04 et 0,8.

Quand $\frac{b}{a} = 2$, les orbites convergent vers un équilibre stable.


```
> #réassignation des paramètres
mu:=0.9;
a,b,c,d:=0.6,1.2,1.04,0.8;
#système dynamique
sdvdp3:=sdvdp;
#représentation graphique
DEplot3d({sdvdp3},[var],t=0..25,[[sigma(0)=1.4,omega(0)=
```

```
0.7,epsilon(0)=0.8],[sigma(0)=2,omega(0)=0.6,epsilon(0)=
0.74]],stepsize=0.01,linecolor=[black,red],orientation=
[86,84],axes=normal,labels=[sigma,omega,epsilon],omega=
0.5..0.75,sigma=1..2.5);
```

```
mu:=0.9
```

```
a,b,c,d:=0.6,1.2,1.04,0.8
```

$$\begin{aligned}
 \text{sdvdp3} &:= \frac{d}{dt} \sigma(t) = \frac{1.666666667 \sigma(t) (1.8 \omega(t) - 1.2)}{1 - \omega(t)}, \quad \frac{d}{dt} \omega(t) \\
 &= \omega(t) \left(1.04 \epsilon(t) - 0.8 - 0.1 \ln \left(\frac{1.8 \omega(t) - 1.2}{1 - \omega(t)} + 1.2 \right) \right), \quad \frac{d}{dt} \epsilon(t) \\
 &= \epsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} - \frac{1.666666667 (1.8 \omega(t) - 1.2)}{1 - \omega(t)} - \ln \left(\frac{1.8 \omega(t) - 1.2}{1 - \omega(t)} \right. \right. \\
 &\quad \left. \left. + 1.2 \right) - 0.01 \right)
 \end{aligned}$$



Par contre, quand $\frac{b}{a} = 3$, la valeur de bifurcation est dépassée, l'équilibre du système dynamique devient instable et toute orbite converge vers un cycle-limite.


```
> a,b:=0.6,1.8;
```

```

sdvdp4:=sdvdp;
DEplot3d({sdvdp4},[var],t=0..50,[[sigma(0)=2,omega(0)=
0.65,epsilon(0)=0.85]],stepsize=0.01,linecolor=[red],
orientation=[176,81,0],axes=normal,labels=[sigma,omega,
epsilon]);

```

$$\begin{aligned}
 & a, b := 0.6, 1.8 \\
 \text{sdvdp4} := & \frac{d}{dt} \sigma(t) = \frac{1.666666667 \sigma(t) (2.4 \omega(t) - 1.8)}{1 - \omega(t)}, \quad \frac{d}{dt} \omega(t) \\
 & = \omega(t) \left(1.04 \varepsilon(t) - 0.8 - 0.1 \ln \left(\frac{2.4 \omega(t) - 1.8}{1 - \omega(t)} + 1.8 \right) \right), \quad \frac{d}{dt} \varepsilon(t) \\
 & = \varepsilon(t) \left(\frac{1 - \omega(t)}{\sigma(t)} - \frac{1.666666667 (2.4 \omega(t) - 1.8)}{1 - \omega(t)} - \ln \left(\frac{2.4 \omega(t) - 1.8}{1 - \omega(t)} \right. \right. \\
 & \left. \left. + 1.8 \right) - 0.01 \right)
 \end{aligned}$$



▼ Références bibliographiques

Shah et Desai, 1981, "Growth Cycles with Induced Technical Change", *Economic Journal*.

Marc Jarsulic, 1986, "Growth Cycles in a Classical-Keynesian Model", in *Competition, Instability and Nonlinear Cycles*, Proceedings of an international conference, New School for Social Research, Lectures notes in Economics and mathematical sysem series, vol 275, Springer.

Peter Skott, 1989, "Effective Demand, Class Struggle and Cyclical Growth", *International Economic Review*, vol. 30, n°1.

Frederik van der Ploeg, 1987, "Growth Cycles, Induced Technical Change, and Perpetual Conflict over the Distribution of Income", *Journal of Macroeconomics*, vol.9, n°1.